


# Struktury počítačových systémů


Přednáší: Richard Šusta

27. října 2011 verze 1.0





ČVUT-FEL v Praze – kód předmětu A0B35SPS

## Overclocking Clocks



<http://www.beige-box.com/>




## DE oscillators


- CLOCK\_27 - 27 MHz clock input ( with TD\_RESET <= '1'; )
- CLOCK\_50 - 50 MHz clock input

The image contains two circuit diagrams for DE oscillators. The left diagram shows a 50MHz oscillator circuit. It features a BC48 crystal connected to a 50MHz oscillator block (Y1). The oscillator block has pins EN, VCC, GND, and OUT. The EN pin is connected to VCC33, the VCC pin is connected to VCC33, the GND pin is connected to ground, and the OUT pin is connected to a 50MHz EXT\_CLOCK input. The right diagram shows a 27MHz oscillator circuit. It features a BC99 crystal connected to a 27MHz oscillator block (Y3). The oscillator block has pins EN, VCC, GND, and OUT. The EN pin is connected to V\_VCC33, the VCC pin is connected to V\_VCC33, the GND pin is connected to ground, and the OUT pin is connected to a 27MHz output. A 1K resistor (R30) is connected to the 27MHz output line.

## Phase Shift and Detector




## Phase Detector with Charge Pump




Eng: Charge pump has 2 strictly different meanings it is a part of phase detector in PLL and also a DC converter (cz:kaskádní násobič)

## Digital VCO Implementation



- Tail current is digitally controlled (for amplitude calibration)
- Cap array size increased (L decreased) to improve tuning range
- Inversion-mode MOS varactor
- No charge pump: Phase detector -> Up/Down Counter

## Basic Block Diagram



$$f_L/M = f_R/N$$

$$f_L = f_R * (M/N)$$

$$f_O = f_R * (M/(N * P))$$

## PLL 27 MHz -> 25,175 Mhz

**25,175MHz**

25,2 MHz -> error < **0.1%**  
 25MHz -> error = -0.7%

27MHz crystal:  
 $10[\text{Hz/C/MHz}] * 27[\text{MHz}]$   
 $* 60[\text{dC}^\circ < -20-80\text{C}^\circ] = \pm 0.06\%$

**ALTPLL**

c0 - Core/External Output Clock  
 Able to implement the requested PLL

☒ Use this clock

**Clock Tap Settings**

☒ Enter output clock frequency: 25.175 MHz (Requested) / 25.200000 (Actual)

☐ Enter output clock parameters:  
 Clock multiplication factor: 14  
 Clock division factor: 15

Clock phase shift: 0.00 deg

Clock duty cycle (%): 50.00

[Less Details <<](#)

Note: The displayed internal settings of the PLL is recommended for use by advanced users only

Description	Value
Primary clock VCO frequency (MHz)	756.000
Modulus for M counter	28
Modulus for N counter	1
Initial VCO phase cycles for M counter	1
VCO phase tap for M counter	0
VCO post scale counter	1

Example:  
 Priority Encoder 16:16 bits

## Truth Table

```

Inputs x(15 downto 0):outputs y(15 downto 0)
1--- ---- ---- ---- : 1000 0000 0000 0000
01-- ---- ---- ---- : 0100 0000 0000 0000
001- ---- ---- ---- : 0010 0000 0000 0000
0001 ---- ---- ---- : 0001 0000 0000 0000
                        * * *
0000 0000 0000 1--- : 0000 0000 0000 1000
0000 0000 0000 01-- : 0000 0000 0000 0100
0000 0000 0000 001- : 0000 0000 0000 0010
0000 0000 0000 0001 : 0000 0000 0000 0001
0000 0000 0000 0000 : 0000 0000 0000 0000
  
```

**Note: Symbol - means don't care**

## Note: Array versus std\_logic\_vector

-- library type

TYPE std\_logic\_vector IS ARRAY ( NATURAL RANGE <>) OF std\_logic;

-- library type

TYPE MojePole IS ARRAY (7 downto 0) of std\_logic;

signal reg : MojePole;

signal reg2 : std\_logic\_vector(7 downto 0);

...

~~reg2<=reg;~~ -- nelze vzájemně převést žádným přetypováním

...


reg2(0)<=reg(0); reg2(1)<=reg(1); -- jediný způsob konverze

VHDL má striktně oddělené typy



## Priority Encoder Entity

```
library ieee; use ieee.std_logic_1164.all;  
entity priority1 is    port (  
    x : in std_logic_vector(15 downto 0);  
    y : out std_logic_vector(15 downto 0) );  
end entity;
```



SPS

15

## Statements used for modeling

### ■ Concurrent statements

- ☐ ...<=.... ;
- ☐ with ... select  
    ...<=... when ...,  
        ...when ...,  
        ...when others;
- ☐ ...<= ... when ... else  
    ... when ... else  
    else ... ;

SPS

16



## With ... select solution

```
architecture dataflow1 of priority1 is
begin with x select
    y <= X"8000" when X"8000" to X"FFFF",
        X"4000" when X"4000" to X"7FFF",
        X"2000" when X"2000" to X"3FFF",
        X"1000" when X"1000" to X"1FFF",
        X"0800" when X"0800" to X"0FFF",
        X"0400" when X"0400" to X"07FF",
        X"0200" when X"0200" to X"03FF",
        X"0100" when X"0100" to X"01FF",
        X"0080" when X"0080" to X"00FF",
        X"0040" when X"0040" to X"007F",
        X"0020" when X"0020" to X"003F",
        X"0010" when X"0010" to X"001F",
        X"0008" when X"0008" to X"000F",
        X"0004" when X"0004" to X"0007",
        X"0002" when X"0002" | X"0003",
        X"0001" when X"0001", X"0000" when others;
end dataflow1;
```

Compile and  
locate priority1  
in Quartus  
RTL Viewer

SPS

17

## Config.vhd

```
configuration config of priority1 is
    for dataflow2
    end for;
end config;
```

for switching  
architectures

SPS

18

## When - else solution

```
architecture dataflow2 of priority1 is
begin
  y <= X"8000" when x(15)='1' else
    X"4000" when x(14)='1' else
    X"2000" when x(13)='1' else
    X"1000" when x(12)='1' else
    X"0800" when x(11)='1' else
    X"0400" when x(10)='1' else
    X"0200" when x(9)='1' else
    X"0100" when x(8)='1' else
    X"0080" when x(7)='1' else
    X"0040" when x(6)='1' else
    X"0020" when x(5)='1' else
    X"0010" when x(4)='1' else
    X"0008" when x(3)='1' else
    X"0004" when x(2)='1' else
    X"0002" when x(1)='1' else
    X"0001" when x(0)='1' else "0000000000000000";
end dataflow2;
```

Compile and  
locate priority1  
again in  
Quartus RTL  
Viewer


SPS

19

## Detector for Snake

```
LIBRARY ieee; USE ieee.std_logic_1164.all;
LIBRARY work;


ENTITY detector IS
  PORT
  (
    sw, had : IN STD_LOGIC_VECTOR(15 DOWNTO 0);
    result : OUT STD_LOGIC
  );
END detector;
```



SPS

20

## Overview of Detector



Convert Symbol to VHDL prototype by  
Create/Update -> Create HDL design file... and edit it.

SPS

21

## VHDL detector

ARCHITECTURE bdf\_type OF detector IS

COMPONENT priority1

PORT(x : IN STD\_LOGIC\_VECTOR(15 DOWNT0 0);

y : OUT STD\_LOGIC\_VECTOR(15 DOWNT0 0)

);

END COMPONENT;

SIGNAL WIRE\_0 : STD\_LOGIC\_VECTOR(15 DOWNT0 0);

BEGIN

b2v\_inst : priority1

PORT MAP(x => sw, y => WIRE\_0);

result <= '0' when (WIRE\_0 and had)= X"0000" else '1';

END bdf\_type;

SPS

22

# Processes

for behavioral modeling

## Process Format

```
[label:] process (sensitivity list)  
    declarations  
begin  
    sequential statements  
end process;
```

- Process allows conventional programming language methods to describe circuit behavior, but the behavior need not always be synthesizable in an ASIC
- Its behavior in a simulation:
  1. Process statements executed once at start of simulation
  2. Process halts at “end” until an event occurs on a signal in the “sensitivity list”
  3. Process statements executed sequentially (sequential statements)

```

architecture behav1 of priority1 is
begin
  process(x)
    variable ix : integer;
    variable enable : boolean;
  begin
    enable:=true; -- inicializations should be in code
    for ix in x'RANGE loop
      if( enable and (x(ix)='1') ) then
        y(ix)<='1'; enable:=false;
      else y(ix)<='0';
      end if;
    end loop;
  end process;
end behav1;

```

Compile and  
locate priority1  
again in  
Quartus RTL  
Viewer

### Wrong inicialization in VHDL

```

architecture behav1 of priority1 is
begin
  process(x)
    variable ix : integer;
    variable enable : boolean:=true;
  begin
    for ix in x'RANGE loop
      if( enable and (x(ix)='1') ) then
        y(ix)<='1'; enable:=false;
      else y(ix)<='0';
      end if;
    end loop;
  end process;
end behav1;

```

← all variable have static durations.  
This code does not work

## Wrong style of behavior


```

architecture behav2 of priority1 is
begin
  process (x)
    variable ix, jx : integer; variable ones : integer;
    begin
      for ix in 15 downto 0 loop
        ones:=0; jx:=ix+1;
        while jx<=15 loop
          if x(jx)='1' then ones:=ones+1; end if;
          jx:=jx+1;
        end loop;
        if ones=0 and x(ix)='1' then y(ix)<= '1'; else y(ix)<='0'; end if;
      end loop;
    end process;
  end behav2;
  
```

Behavior explanation too complex for RTL

Compile and locate *priority1* again in Quartus RTL Viewer to see why it is terrible style

27





## Variable assignment := execution

---

- Variables can only be declared and used in the sequential part (inside process or function) of VHDL local to a process.
- variable := variable assignment.
- A2 := B2 and C2
- similar to signal assignment but A2 must be a variable.

## Latches and Flip-flops

In VHDL

## Latches

*Latches can be easily described by using the concurrent signal assignment statement*

```
library IEEE;
use IEEE.STD_LOGIC_1164.all;
```

```
entity JK_LATCH is
port ( J, K : in std_logic;
      signal Q : out std_logic);
end JK_LATCH;
```

```
architecture TRUTH_TABLE of JK_LATCH is
signal Q_MEM : std_logic;
begin
```

*-- Map truth table into conditional concurrent statements*

```
Q_MEM <= Q_MEM when (J = '0' and K = '0') else
  '0' when (J = '0' and K = '1') else
  '1' when (J = '1' and K = '0') else
  not Q_MEM;
```

```
Q <= Q_MEM;
end TRUTH_TABLE;
```

*Hence, the JK latch is identical to an SR latch that is made to toggle its output when passed the restricted combination*

J	K	Q+
0	0	Q
0	1	0
1	0	1
1	1	Q'

SPS

## Gated D-Latch

```
library IEEE;
use IEEE.STD_LOGIC_1164.all;
```

```
entity latch_ex is
port (gate, in1 : in std_logic;
      out1 : out std_logic);
end latch_ex;
```


```
architecture latch_ex_arch of latch_ex is
begin
  process (gate, in1)
  begin
    if (gate = '1') then
      out1 <= in1;
    end if;
  end process;
end latch_ex_arch;
```

### Latch

1-bit memory

in1 → D Q → out1

gate → C





### sensitivity list

- the process executes once when the gate or in1


SPS

32






## Detecting Edges




## Flip-flop with asynchronous reset

```
architecture dff_asyn_arch of dff_asyn is
begin
  process(clock, reset)
  begin
    if (reset = '1') then
      out1 <= '0';
      --rising_edge(clock)
    elsif clock = '1' and clock\'event then
      out1 <= in1;
    end if;
  end process;
end dff_asyn_arch;
```



## Miniotázky: nakreslete ql, qe



Co se stane, při náběžné hraně clock, pokud bude reset='1' ?

```

1 architecture dff_asyn_arch of dff_asyn is
2 begin
3   process(clock, reset)
4   begin
5     if (reset = '1') then
6       out1 <= '0';
7     elsif clock = '1' and clock'event then
8       out1 <= in1;
9     end if;
10  end process;
11 end dff_asyn_arch;
```


clock and reset  
must be in  
sensitivity list

## Flip-flop with synchronous reset

```

architecture dff_syn_arch of dff_syn is
begin process(clock,reset) -- reset can be removed,
begin-- -- but allowed
    if clock = '1' and clock'event then
        if (reset = '1') then
            out1 <= '0';
        else
            out1 <= in1;
        end if;
    end if;
end process;
end dff_syn_arch;

```



Why reset is not needed in the sensitivity list?

SPS

39

## Difference between Syn. & Asyn. flip-flops

*The order of the statements inside the process determines Syn. or Asyn. reset*

```

if clock = '1' and clock'event then
    if (reset = '1') then

```

1

```

if (reset = '1') then

```

2

```

    q <= '0';

```

```

elsif clock = '1' and clock'event then

```

*Which flip-flop has synchronous and asynchronous reset?*

SPS

40

## Miniquestions

- What is the difference between latches and flip-flops (level triggered flip-flops and edge triggered flip-flops)?

Note: \*\*In some FPGA all flip-flops are treated as 50% edge triggered flip-flops.

- What is the difference between
  - synchronous reset (syn-reset) flip-flops and
  - asynchronous reset (asyn-reset) flip-flops?



SPS

41

Example:  
Divider by 50 milions  
with 50% duty cycle



SPS

42

## Synchronous Divider

```
-- pridame delic 2 k delici 25000000 -> delic50000000
library ieee; use ieee.std_logic_1164.all; use
    ieee.std_logic_unsigned.all;
-- delic 50e6 se symetrickym vystupem
entity delic50M is
    port(clk : in std_logic;
          q : out std_logic);
end entity;

architecture behav1 of delic50M is
    constant MAX:integer :=24999999;
begin
    process (clk)
        --processes body on the next slide
    end process;
end behav1;
```



SPS

43

## Synchronous Divider

```
process (clk)
    variable delic2 : std_logic := '0';
    variable cnt : integer:=0;
begin
    if (clk'event and clk='1') then
        if cnt=MAX then
            cnt:=0; delic2:=not delic2;
        else cnt := cnt + 1;
        end if;
    end if;
    q<=delic2;
end process;
```

**Too high gate count**  
**56 LE / 33 Reg**  
 LE=Logic elements  
 Reg - registers



SPS

44

## Information


- If we supply detailed information about circuit limit VHDL compiler will create more optimal circuit
  - Is it required 32 bit cnt variable ? -> **No**
  - Is it necessary perform equality comparison? -> **No**



SPS

45

## Improved Behavior Description


```
process (clk)
  variable delic2 : std_logic := '0';
  variable cnt : integer range 0 to MAX := 0;
begin
  if (clk'event and clk='1') then
    if cnt=MAX then
      cnt:=0; delic2:=not delic2;
    else cnt := cnt + 1;
    end if;
  end if;
  q<=delic2; 
end process;
```

46LE/26Reg  
by adding **range 0 to MAX** information we simplified compilation

SPS

46

## Remeber Two-Level Simplification



*Equality is expensive operation*

$$F_1 = A'B'C'D' + A'BC'D + ABCD + AB'C'D'$$

$$F_2 = A'B'D + A'C$$

$$F_3 = BC'D' + AC' + ABD'$$

$$F_1 = F_2' \cdot F_3'$$

SPS



47

## Synchronous Divider

```

process (clk)
  variable delic2 : std_logic := '0';
  variable cnt : integer range 0 to MAX := 0;
begin
  if (clk'event and clk='1') then
    if cnt >= MAX then
      cnt := 0; delic2 := not delic2;
    else cnt := cnt + 1;
    end if;
  end if;
  q <= delic2;
end process;

```


37LE/26Reg  
compare by >=  
we added another  
information

SPS

48



## Synchronous Divider

```
process (clk)
  variable delic2 : std_logic := '0';
  variable cnt : integer range 0 to MAX := 0;
begin
  if (clk'event and clk='1') then
    if cnt > MAX-1 then ..... 36LE/26Reg
      cnt:=0; delic2:=not delic2; compare by >
    else cnt := cnt + 1;
    end if;
  end if;
  q<=delic2; 
end process;
```

SPS

49


## Information

By inserting detailed information  
about circuit behavior,  
**we have 36LE/26Reg**  
i.e. we saved 20LE/9Reg  
from original 56LE/33Reg

SPS

50

## Universal divider

```
library ieee;use ieee.std_logic_1164.all;use ieee.std_logic_unsigned.all;
entity FreqDivByEven is
  generic( Even_Divisor: integer := 50 );
  port(CLK : in std_logic;
        q : out std_logic);
end entity;
architecture behav of FreqDivByEven is
  constant Divider:integer :=Even_Divisor/2;
begin
  -- FreqDivByEven requires EVEN number
  assert Even_Divisor mod 2=0 -- tested in compile time
  report "FreqDivByEven requires EVEN divisor"
    severity severity_level(error);
  process (CLK)
     see the next slide
  end process;
end behav;
```

SPS

51

## Universal divider

```
process (CLK)
  variable cnt : integer range 0 to Divider-1:=0;
  variable q2 : std_logic := '0';
begin
  if (CLK'event and CLK='1') then
    if Even_Divisor=2 then q2:=not q2;
    elsif cnt>Divider-2 then cnt:=0; q2:=not q2;
    else cnt := cnt + 1;
    end if;
  end if;
  q<=q2;
end process;
```

SPS

52

## One divider with parameter 50milions

Quartus info for divider by 50 milions  
created as one circuit:


- **36LE/26Reg**
- Info: Clock "clk" has Internal fmax of **263.99** MHz between source register



SPS

53

## The better divider by 50 milions



**Gate count only 33 LE/ 26Reg**

Info: Clock "clk" has Internal fmax of **338.29 MHz** between source register "FreqDivByEven:inst1|cnt[0]" and destination register "FreqDivByEven:inst1|q2" (period= 2.956 ns)

SPS

54

# Decomposition

We can obtain more optimal and effective circuit by decomposing huge designs to several smaller units

- e.g. by proper splitting one 50 million divider to 3 dividers 50,1000, and 1000

